Problem:
Score : 400 points

Problem Statement

One day, Niwango-kun, an employee of Dwango Co., Ltd., found an integer sequence (a_1, ..., a_N) of length N.
He is interested in properties of the sequence a.

For a nonempty contiguous subsequence a_l, ..., a_r (1 \leq l \leq r \leq N) of the sequence a, its beauty is defined as a_l + ... + a_r. Niwango-kun wants to know the maximum possible value of the bitwise AND of the beauties of K nonempty contiguous subsequences among all N(N+1)/2 nonempty contiguous subsequences. (Subsequences may share elements.)

Find the maximum possible value for him.

Constraints

2 \leq N \leq 1000

1 \leq a_i \leq 10^9

1 \leq K \leq N(N+1)/2

All numbers given in input are integers

Input

Input is given from Standard Input in the following format:

N K
a_1 a_2 ... a_N

Output

Print the answer.

Sample Input 1

4 2
2 5 2 5

Sample Output 1

12

There are 10 nonempty contiguous subsequences of a. Let us enumerate them:

contiguous subsequences starting from the first element: \{2\}, \{2, 5\}, \{2, 5, 2\}, \{2, 5, 2, 5\}

contiguous subsequences starting from the second element: \{5\}, \{5, 2\}, \{5, 2, 5\}

contiguous subsequences starting from the third element: \{2\}, \{2, 5\}

contiguous subsequences starting from the fourth element: \{5\}

(Note that even if the elements of subsequences are equal, subsequences that have different starting indices are considered to be different.)

The maximum possible bitwise AND of the beauties of two different contiguous subsequences is 12.
This can be achieved by choosing \{5, 2, 5\} (with beauty 12) and \{2, 5, 2, 5\} (with beauty 14).

Sample Input 2

8 4
9 1 8 2 7 5 6 4

Sample Output 2

32

Solution in Haskell (GHC 7.10.3) (judge verdict: Accepted)

import           Control.Monad (guard)
import           Data.Bits     (bit, (.&.))
import           Data.List     (inits, mapAccumL, tails, unfoldr)
import           Data.Tuple    (swap)

main :: IO ()
main = do
  [n, k] <- take 2 . map read . words <$> getLine
  as <- take n . map read . words <$> getLine
  print $ f k as

f :: Int -> [Int] -> Int
f k as = sum $ snd $ mapAccumL g beautifulnesses bits
  where
    beautifulnesses = map sum . concatMap (tail . inits) . init . tails $ as
    bits = reverse $ map bit [0 .. bitSize (maximum beautifulnesses)]
    g bs i =
      let
        bs' = filter ((== i) . (.&. i)) bs
      in
        if length bs' >= k then (bs', i) else (bs, 0)

bitSize :: Int -> Int
bitSize = length . toDigits 2

toDigits :: Integral a => a -> a -> [a]
toDigits base = reverse . unfoldr step
  where
    step n = do
      guard $ n /= 0
      return $ swap $ divMod n base